check Convergence order of numeric scheme

simple wave

In[◦]:= `Quit[]`

In[◦]:= `wave = 1 / c ^ 2 D[F[t, r], {t, 2}] - D[D[F[t, r], r], r] - S[t, r] // Simplify`

Out[◦]= $$-S[t, r] - F^{(0,2)}[t, r] + \frac{F^{(2,0)}[t, r]}{c^2}$$

In[◦]:= `p = c dt / dr;`

In[◦]:= `eq[n_, i_] =`
`(-(u[n + 1, i + 1] + u[n + 1, i - 1]) + 2 (1 + 2 / p ^ 2) u[n + 1, i] - 2 (u[n, i + 1] + u[n, i - 1] - 2 (1 - 2 / p ^ 2) u[n, i]) -`
`(u[n - 1, i + 1] + u[n - 1, i - 1] - 2 (1 + 2 / p ^ 2) u[n - 1, i]) - 4 c ^ 2 dt ^ 2 / p ^ 2 σ[n, i]) / dr ^ 2`

Out[◦]= $$\frac{1}{dr^2}\left(-u[-1+n, -1+i] + 2\left(1+\frac{2\,dr^2}{c^2\,dt^2}\right)u[-1+n, i] - u[-1+n, 1+i] - 2\left(u[n, -1+i] - 2\left(1-\frac{2\,dr^2}{c^2\,dt^2}\right)u[n, i] + u[n, 1+i]\right) - u[1+n, -1+i] + 2\left(1+\frac{2\,dr^2}{c^2\,dt^2}\right)u[1+n, i] - u[1+n, 1+i] - 4\,dr^2\,\sigma[n, i]\right)$$

In[◦]:= `(*r[i_]=i dr;*)`

In[◦]:= `F[t_, r_] = (Series[f[t0 + ϵ (t - t0), r0 + ϵ (r - r0)], {ϵ, 0, 5}] // Normal) /. ϵ → 1;`
`S[t_, r_] = (Series[s[t0 + ϵ (t - t0), r0 + ϵ (r - r0)], {ϵ, 0, 5}] // Normal) /. ϵ → 1;`

In[◦]:= `wave0 = wave /. {r → r0, t → t0}`

Out[◦]= $$-s[t0, r0] - f^{(0,2)}[t0, r0] + \frac{f^{(2,0)}[t0, r0]}{c^2}$$

In[◦]:= `sol = Solve[wave0 == 0, f^(2,0)[t0, r0]] // Flatten`

Out[◦]= $$\left\{f^{(2,0)}[t0, r0] \to c^2\left(s[t0, r0] + f^{(0,2)}[t0, r0]\right)\right\}$$

In[◦]:= `u[n_, i_] := F[t0 + n dt, r0 + i dr]`
`σ[n_, i_] := S[t0 + n dt, r0 + i dr]`

In[◦]:= `((eq[0, 0] /. {r[0] → r0, dr → ϵ dr, dt → ϵ dt})) // Simplify`

Out[◦]= $$-4\,s[t0, r0] - 4\,f^{(0,2)}[t0, r0] - \frac{1}{3}\,dr^2\,\epsilon^2\,f^{(0,4)}[t0, r0] + \frac{4\,f^{(2,0)}[t0, r0]}{c^2} - dt^2\,\epsilon^2\,f^{(2,2)}[t0, r0] + \frac{dt^2\,\epsilon^2\,f^{(4,0)}[t0, r0]}{3\,c^2}$$

In[∘]:=
```
((eq[0, 0] /. {r[0] → r0, dr → ϵ dr, dt → ϵ dt}) + O[ϵ]^4) /. sol // Simplify
```

Out[∘]=

$$\frac{1}{3}\left(-dr^2\, f^{(0,4)}[t0,\ r0] + dt^2\left(-3\, f^{(2,2)}[t0,\ r0] + \frac{f^{(4,0)}[t0,\ r0]}{c^2}\right)\right)\epsilon^2 + O[\epsilon]^4$$

Wave in Spherical Symmetry

In[∘]:=
```
Quit[]
```

In[1]:=
```
wave = 1 / c^2 D[F[t, r], {t, 2}] - 1 / r^2 D[r^2 D[F[t, r], r], r] - S[t, r] // Simplify
```

Out[1]=

$$-S[t,\ r] - \frac{2\, F^{(0,1)}[t,\ r]}{r} - F^{(0,2)}[t,\ r] + \frac{F^{(2,0)}[t,\ r]}{c^2}$$

In[6]:=
```
p = c dt / dr;
```

In[7]:=
```
eq[n_, i_] =
 (-(u[n + 1, i + 1] + u[n + 1, i - 1]) + 2 (1 + 2 / p^2) u[n + 1, i] - 2 (u[n, i + 1] + u[n, i - 1] - 2 (1 - 2 / p^2) u[n, i]) -
    (u[n - 1, i + 1] + u[n - 1, i - 1] - 2 (1 + 2 / p^2) u[n - 1, i]) -
    4 c^2 dt^2 / p^2 (2 / r[i] (u[n, i + 1] - u[n, i - 1]) / (2 dr) + σ[n, i])) / dr^2
```

Out[7]=

$$\frac{1}{dr^2}\left(-u[-1+n,\ -1+i] + 2\left(1+\frac{2\,dr^2}{c^2\,dt^2}\right)u[-1+n,\ i] - u[-1+n,\ 1+i] - 2\left(u[n,\ -1+i] - 2\left(1-\frac{2\,dr^2}{c^2\,dt^2}\right)u[n,\ i] + u[n,\ 1+i]\right) - u[1+n,\ -1+i] + 2\left(1+\frac{2\,dr^2}{c^2\,dt^2}\right)u[1+n,\ i] - u[1+n,\ 1+i] - 4\,dr^2\left(\frac{-u[n,\ -1+i]+u[n,\ 1+i]}{dr\ r[i]} + \sigma[n,\ i]\right)\right)$$

In[∘]:=
```
(*r[i_]=i dr;*)
```

In[8]:=
```
F[t_, r_] = (Series[f[t0 + ϵ (t - t0), r0 + ϵ (r - r0)], {ϵ, 0, 5}] // Normal) /. ϵ → 1;
S[t_, r_] = (Series[s[t0 + ϵ (t - t0), r0 + ϵ (r - r0)], {ϵ, 0, 5}] // Normal) /. ϵ → 1;
```

In[10]:=
```
wave0 = wave /. {r → r0, t → t0}
```

Out[10]=

$$-s[t0,\ r0] - \frac{2\, f^{(0,1)}[t0,\ r0]}{r0} - f^{(0,2)}[t0,\ r0] + \frac{f^{(2,0)}[t0,\ r0]}{c^2}$$

In[11]:=
```
sol = Solve[wave0 == 0, f^(2,0)[t0, r0]] // Flatten
```

Out[11]=

$$\left\{f^{(2,0)}[t0,\ r0] \to \frac{c^2\left(r0\ s[t0,\ r0] + 2\, f^{(0,1)}[t0,\ r0] + r0\ f^{(0,2)}[t0,\ r0]\right)}{r0}\right\}$$

In[12]:=
```
u[n_, i_] := F[t0 + n dt, r0 + i dr]
σ[n_, i_] := S[t0 + n dt, r0 + i dr]
```

In[14]:=

```
((eq[0, 0] /. {r[0] → r0, dr → ϵ dr, dt → ϵ dt})) // Simplify
```

Out[14]=

$$-4\, s[t0,\, r0] - \frac{8\, f^{(0,1)}[t0,\, r0]}{r0} - 4\, f^{(0,2)}[t0,\, r0] - \frac{4\, dr^2\, \epsilon^2\, f^{(0,3)}[t0,\, r0]}{3\, r0} - \frac{1}{3}\, dr^2\, \epsilon^2\, f^{(0,4)}[t0,\, r0] - \frac{dr^4\, \epsilon^4\, f^{(0,5)}[t0,\, r0]}{15\, r0} + \frac{4\, f^{(2,0)}[t0,\, r0]}{c^2} - dt^2\, \epsilon^2\, f^{(2,2)}[t0,\, r0] + \frac{dt^2\, \epsilon^2\, f^{(4,0)}[t0,\, r0]}{3\, c^2}$$

In[15]:=

```
((eq[0, 0] /. {r[0] → r0, dr → ϵ dr, dt → ϵ dt}) + O[ϵ]^4) /. sol // Simplify
```

Out[15]=

$$\frac{1}{3}\left(-\frac{4\, dr^2\, f^{(0,3)}[t0,\, r0]}{r0} - dr^2\, f^{(0,4)}[t0,\, r0] + dt^2\left(-3\, f^{(2,2)}[t0,\, r0] + \frac{f^{(4,0)}[t0,\, r0]}{c^2}\right)\right)\epsilon^2 + O[\epsilon]^4$$